TP6

Polynômes et fractions rationnelles

1 Polynômes

Un polynôme est une expression dépendant d'une ou plusieurs variables, ces variables intervenant dans l'expression sous forme de produits et de sommes. Une telle expression est de type `polynom`.

> `A:=X^2+X+1; type(A,polynom);`

$$A:=X^2+X+1$$
true

Pour évaluer le polynôme A en une valeur α, il faut utiliser la fonction `subs`. Par exemple, pour calculer $A(5)$,

> `subs(X=5,A);`

31

Il existe des fonctions qui permettent d'isoler les termes d'un polynôme P d'indéterminée X :

`>coeff(P,X,i);`	Donne le coefficient de P du monôme de degré i.
`>coeffs(P,X);`	Donne les coefficients de P sous forme d'une séquence.
`>degree(P);`	Donne le degré du polynôme P.
`>lcoeff(P);`	Donne le coefficient dominant du polynôme P.
`>ldegree(P);`	Donne le degré du terme de plus bas degré du polynôme P.
`>tcoeff(P);`	Donne le coefficient du terme de plus bas degré du polynôme P.

On peut aussi développer et factoriser les polynômes :

> `B:=(X-3)*(X-2);`

B:=(X-3)(X-2)

Par défaut, Maple ne développe donc pas les expressions. Voici des fonctions qui permettent de développer un polynôme, de le factoriser, mais aussi de modifier l'expression d'un polynôme :

`>expand(P);`	Développe le polynôme P.
`>factor(P);`	Factorise le polynôme P dans le corps des rationnels.
`>factor(P,complex);`	Factorise le polynôme P dans le corps des complexes.
`>collect(P,X);`	Regroupe les termes de même puissance en X dans l'expression du polynôme P.
`>collect(P,X,factor);`	Idem ci-dessus en factorisant les coefficients.
`>sort(P,X);`	Ordonne le polynôme P suivant les puissances décroissantes de X.

Seule la dernière instruction modifie effectivement l'écriture du polynôme en mémoire. Par exemple, pour la factorisation, `factor(P)` écrit le polynôme P sous forme factorisé mais il n'est pas modifié. Pour modifier le polynôme, il faut utiliser la commande `P:=factor(P)`.
On peut aussi tester la divisibilité entre deux polynômes avec la fonction `divide`. Par exemple, pour les polynômes A et B :

> `divide(A,B,X);`

false

Voici pour terminer une liste de fonctions pour calculer le quotient, le reste, le pgcd... de deux polynômes :

`>quo(P,Q,X);`	Calcule le quotient de la division euclidienne de P et Q.
`>rem(P,Q,X);`	Calcule le reste de la division euclidienne de P et Q.
`>gcd(P,Q,X);`	Calcule le pgcd des polynôme P et Q.
`>lcm(P,Q,X);`	Calcule le ppcm des polynôme P et Q.
`` >gcdex(P,Q,X,`U`,`V`); ``	Calcule le pgcd et les polynômes de Bezout des polynômes P et Q.

Pour cette dernière instruction, les polynômes de Bezout sont affectés aux variables U et V. Par exemple :

> `` gcdex(A,B,X,`U`,`V`); ``

```
> U,V;
```

$$\frac{25}{91}-\frac{6}{91}X, \frac{11}{91}+\frac{6}{91}X$$

```
> simplify(U*A+V*B);
```

$$1$$

2 Fractions rationnelles

Rappelons qu'une fraction rationnelle est de la forme : $\dfrac{a_nX^n+a_{n-1}X^{n-1}+\ldots+a_1X+a_0}{b_mX^m+b_{m-1}X^{m-1}+\ldots+b_1X+b_0}$. Une telle expression est du type `ratpoly`.

```
> F:=(X^4+3*X^3+2*X^2+X+1)/(X^3+2*X^2+2*X+1);
```

$$F := \frac{X^4+3X^3+2X^2+X+1}{X^3+2X^2+2X+1}$$

```
> type(F,ratpoly);
```

true

Comme pour les polynômes, on peut évaluer une fraction rationnelle en une valeur α grâce à la fonction `subs`. On peut extraire d'une fraction rationnelle son numérateur ou son dénominateur avec les fonctions `numer` et `denom`. La fonction `normal` permet de simplifier la fraction rationnelle. Par exemple :

```
> normal(F);
```

$$\frac{X^3+2X^2+1}{X^2+X+1}$$

La fonction `factor` peut elle aussi être utilisée pour simplifier et factoriser une fraction rationnelle. Ces deux dernières fonctions ne modifient pas l'expression de la fraction rationnelle en mémoire. Enfin, il existe une fonction `convert` qui permet de décomposer en éléments simples une fraction rationnelle à priori dans $\mathbb{Q}$ ou le sur-corps engendré par ces coefficients.

```
> convert(F,parfrac,X);
```

$$X+1-\frac{2X}{X^2+X+1}$$

Exercice 1

1. Factoriser dans $\mathbb{R}$ puis dans $\mathbb{C}$ le polynôme suivant : $X^4+3X^3-4X^2-5X-6$.
2. Développer et ordonner l'expression suivante : $(X-3)(X^2+1)(X+1)$.
3. Trouver a,b,c tels que le polynôme $X^6+\sqrt{2}X^5+aX^2+bX+c$ ait une racine d'ordre $\geq$ 4.
4. Déterminer le quotient et le reste de la division euclidienne de $X^{12}+8X^{11}+5X^6-3X^4+X^2-5$ par X^3-1.
5. Déterminer le pgcd de $A=X^7-X-1$ et $B=X^5+1$, et les polynômes U et V tels que $AU+BV=A\wedge B$.

Exercice 2

1. Simplifier et décomposer en éléments simples la fraction rationnelle $\dfrac{X+2}{X^4+X^3-2X^2}$.
2. On considère la fraction rationnelle $F=\dfrac{X^5-15X^4+85X^3-225X^2+274X-120}{X^6+21X^5+175X^4+735X^3+1624X^2+1764X+720}$.
   (a) Factoriser F.
   (b) Calculer, avec la décomposition en éléments simples de F, l'unique 6-uplet de réels $(p_0,\ldots,p_5)$ tel que $F=\displaystyle\sum_{k=0}^{5}\frac{p_k}{X+k+1}$.
   (c) Soit $P(X)=p_0+p_1X+\ldots+p_nX^n$. Calculer, pour $i=0,\ldots,5$, $\int_0^1 P(t)t^i dt$ et interpréter le résultat.

Exercice 3 Soit à résoudre dans $\mathbb{R}[X]$ l'équation $(\mathcal{E})$: $(X^2+1)P'' = 6P$. Il est facile de constater que le degré d'un polynôme P non nul qui satisfait à l'équation $(\mathcal{E})$ est 3 (considérer les coefficients dominants). Une méthode de résolution peut consister à :

- créer un polynôme P de degré au plus 3,
- former le polynôme $(X^2+1)P''-6P$, l'écrire sous forme réduite,
- créer la séquence S des équations traduisant la nullité du polynôme $(X^2+1)P''-6P$,
- résoudre le système des quatre équations obtenues,
- affecter les valeurs des solutions aux coefficients de P,
- et enfin afficher P.

Pour cela, on pourra utiliser les commandes suivantes :

```
> restart:  P:=sum(a[k]*x^k,k=0..3);
  Q:=(x^2+1)*diff(P,x$2)-6*P;
  Q:=collect(Q,x);
  S:=coeffs(Q,x);
  solve({S});
  assign(%);
  P;
```

Tester cette méthode, puis l'adapter à la résolution des quatre problèmes qui suivent :

1. Résoudre dans $\mathbb{R}[X]$ l'équation : $(P')^2 = 4P$.
2. Résoudre dans $\mathbb{R}[X]$ l'équation : $(X^5-1)P^{(5)} = 120P$.

Exercice 4 Écrire une procédure `Tcheby` qui permet de déterminer T_n, où T_n représente le n-ième polynôme de Tchebychev. Rappelons que les polynômes de Tchebychev sont définis par la relation de récurrence :

$$T_0(x) = 1,\ T_1(x) = x \text{ et } T_n(x) = 2xT_{n-1}(x) - T_{n-2}(x).$$

Signalons au passage que les fonctions du package `orthopoly` donnent accès aux polynômes orthogonaux classiques, et en particulier aux polynômes de Tchebychev. On pourra ainsi vérifier la validité de la procédure `Tcheby`.

Exercice 5 Étant donné un polynôme $P = a_0 + a_1X + \ldots + a_nX^n$, et $x \in \mathbb{R}$, il s'agit de calculer $P(x)$.

1. Combien faut-il d'opération pour évaluer $P(x)$ "directement"?
2. L'algorithme de Hörner consiste à écrire :

$$P(X) = a_0 + X(a_1 + X(a_2 + \ldots + X(a_{n-1} + a_nX)\ldots)).$$

   Combien faut-il d'opérations avec cette écriture? Écrire une procédure `horner` qui, étant donné un polynôme P et $x \in \mathbb{R}$, calcule $P(x)$ en utilisant la factorisation précédente.